
Kapittel 1

Opplegg

1.1 Undervisningsopplegg

Dette kompendiet dekker pensum i fagetTE 6192 Reservoarsimulering, innføringskurs
ved Institutt for petroleumsteknologi. Undervisningen er praktisk lagt opp med øvinger
på datamaskin.

Etter en kortfattet gjennomgåelse av det teoretiske grunnlaget blir det utdelt et
fortran -program som simulerer enfasestrøm i en dimensjon. Programkoden blir
gjennomgått systematisk. På dette grunnlag skal hver student utvikle sin egen simu-
leringsmodell ved å utvide den utdelte modellen gjennom øvingsoppgavene. Noen av
oppgavene vil være obligatoriske og skal leveres inn for godkjenning.

I de første ukene blir det forelesninger i 4 timer per uke, inntil det teoretiske grunn-
laget er gjennomgått. Deretter vil det bli delt ut en ny øving hver uke. Denne blir
gjennomgått med forslag og hint til løsning i løpet av 2 forelesningstimer. De resteren-
de to undervisningstimene benyttes til veiledning på terminalrommet. Det reserveres
tid på datarommet for hver student tilsvarende en formiddag per uke, men ellers er det
åpent for den enkelte å benytte datamaskinene etter behov.

1.2 Faglig omfang

Det tas utgangspunkt i en simuleringsmodell (etfortran -program) som løser nume-
risk den differensialligningen som styrer horisontal strøm av en fase i en dimensjon.
Reservoaret har uniforme egenskaper, og den numeriske formuleringen som løser dif-
ferensialligningen er av enkleste slag.Del I av kompendiet omhandler det teoretiske
grunnlag for denne ligningen. IDel II inngår fortran -koden med inngangsfil og
resultatfil. Følgende emner vil bli behandlet på denne måten, så langt tiden rekker:

1. Utprøving av utlevert program.

2. Automatisk tidsstegsregulering

KAPITTEL 1. OPPLEGG 3

3. Sammenligning med analytisk løsning.

4. Ulike numeriske formuleringer og løsningsmetoder.

5. Varierende blokklengde, permeabilitet og porøsistet.

6. Hellende reservoar.

7. Radiell strøm.

8. Tabellbruk.

9. Strøm av en fase i to dimensjoner.

10. Strøm av to faser i en dimensjon.

11. Måling av relative permeabiliteter.

På siden http://www.ux.his.no/�s-skj/ vil en finne en peker til faget med blant annet
Siste Nytt, eksamensoppgaver og utkast til noen løsninger, samt annen informasjon.

1.3 Litteratur

Som støttebok anbefales boken til Aziz og Settari [1]. Det kan ellers kanskje svare seg
å ha tilgang på en håndbok ifortran90 og eventuelt lærebøker i numeriske metoder,
selv om dette neppe er nødvendig. Se ellers pekere på hjemmesiden til faget.

På biblioteket er det konferanserartikler fra de siste symposiene i numerisk re-
servoarsimulering. Der vil en få inntrykk av hvilke problemstillinger som er aktuelle
innen denne grenen av petroleumsteknologi. Dessuten er det publisert mange simule-
ringsstudier i den ordinære petroleumslitteraturen.Society of Petroleum Engineers har
stått for utabeidelsen av en monografi om reservoarsimulering. Redaktørene Mattax
og Dalton [2] sammen med andre bidragsytere har gitt et godt sammendrag av hva
reservoarsimulering dreier seg om.

1.4 Fontkonvensjon

Et kompendium som dette bør i hovedsak holde seg til tre typer fonter. Vanlig tekst
skrives som dette. Latinske bokstaver som brukes som matematiske symboler skrives
med kursiv:a;b;c. I tillegg vil vi få bruk for å angifortran -kode. Variable og utsagn
i FORTRANskrives med skrivemaskintyper som følger:if (oxmin .lt. 0.d0)
oxmin = 0.d0 .

det er forsøkt å velgefortran variabelnavn som er lette å huske, som ligner på
det matematiske symbolet for samme variable. for eksempel brukespo for oljetrykket
po.

1.5 Fortranversjon

Kurset gjennomføres i Fortran90 (F90) som har F77 som en ekte delmengde. Program-
varen som blir brukt er Compaque Visual Fortran. Programmet har god hjelpefunksjon
for språket generelt og for selve programbruken. Andre innføringsopplegg i F90 vil
ligge utlagt på http://www.ux.his.no/�s-skj/. En bør være oppmerksom på at flere av
kodeforslagene kan innholde rester etter F77 og det er ikke sikkert at alle vil virke uten
videre.



Kapittel 2

Diffusivitetsligningen

Vi skal utlede diffusivitetsligningen på to litt ulike måter, først ved å starte med Darcy’s
lov og massebevarelse, og deretter mer direkte fra den generelle diffusivitetsligning.

Vi begrenser oss til såkaltbeta-formulering. Det vil si at fluidene kan beskrives
ved hjelp av de tradisjonelle volumfaktoreneBo, Rso, Bg, Rsg, Bw, som alle kun er
funksjoner av trykket. Dette medfører at modellen er lite egnet til å simulere prosesser i
reservoaret dersom det er store endringer i sammensetningen av fasene. Uttrykketbeta-
formulering kommer av at volumfaktorene tidligere ble betegnet med gresk bokstavβ .

Vi skal starte med lineær, horisontal strøm av en fase i en dimensjon i et reservoar
med konstant porøsitet og permeabilitet.

2.1 Darcy’s lov og massebevarelse

Darcy’s lov kan uttrykkes på følgende måte:

uo =�
Ckx

µo

∂ po

∂x
; . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . (2.1)

hvor

C : Omregningskonstant med verdi 0.00632827 for de gitte enhetene.

uo : volumhastigheten, ft/d.

kx : absolutt permeabilitet ix-retning, md.

µo : viskositet, cp.

po : trykk, psi.

x : avstand langs strømningsretningen, ft.

KAPITTEL 2. DIFFUSIVITETSLIGNINGEN 6

Indekso står for olje og er egentlig overflødig før vi tar med strøm av flere faser.
Kontinuitetsligningen kan utledes ved å betrakte massestrøm inn og ut av volumele-

mentet∆V = A∆x i Figur 2.1.

Av0ρ0 Av0ρ0+∆(Av0ρ0)
A

∆x

q0∆xAρ0/b0

Figur 2.1: Masserate inn og ut av volumelementA∆x.

I Figur 2.1 er det brukt endel nye størrelser:

A : tverrsnitt av det en-dimensjonale reservoaret, ft2.

qo : oljerate per bulkvolum, stb/d/bbl. Raten er positiv for injeksjon og negativ for
produksjon. Dette er kildeleddet i kontinuitetsligningen [3].1

bo : volumfaktor for olje, stb/rb.2

ρo : tetthet av olje ved reservoarbetingelser, lb/rft3.

Endring av masse per tidsenhet i volumelementet∆V på grunn av strøm inn eller
ut, samt injeksjon eller produksjon er gitt ved

A(ρouo +
qo∆xρo

bo
� (uoρo +∆(uoρo))): . . . . . . . . . . . . . . . . . . (2.2)

Masse tilstede i volumelementet erA∆xφρo og endring av denne massen per tidsenhet
blir

∂
∂ t

(A∆xφρo): . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . (2.3)

1Merk at bbl her betegner volumenheten ’fat’ av bergartsvolum.
2Merk at denne definisjonen er den inverse av den vanlige volumfaktorenBo og at rb betegner “re-

servoir barrel” av fluid.

Settes uttrykket 2.2 lik uttrykket 2.3, og antas porøsitetenφ konstant, får en etter
litt forenkling

∆xφ
∂ρo

∂ t
=�∆(uoρo)+

qo∆xρo

bo
:

Vi deler nå ligningen med∆x og lar∆x! 0, og får

φ
∂ρo

∂ t
=�

∂ (ρouo)

∂x
+

qoρo

bo
:

Denne ligningen multipliserer vi medbo=ρo som har dimensjonen stb/rb� rft3/lb !
5:61�stb/lb. Dette er en konstant og kan tas innenfor derivasjonstegnet:

φ
∂
�

ρo
bo

ρo

�

∂ t
=�

∂
�

ρo
bo

ρo
uo

�

∂x
+

qoρo

bo

bo

ρo
;

som ordnet gir

∂ (bouo)

∂x
�qo =�φ

∂bo

∂ t
: . . . . . . . . . . . . . . . . . . . . . . . . . . (2.4)

Ligning 2.4 er kontinuitetsligningen skrevet på en noe spesiell form med volumfak-
tor istedenfor tetthet. Det er imidlertid den vanlige framstillingsmåten forβ–modeller.
Ligningen uttrykker bevarelse av overflatevolum av olje. Dette er det samme som mas-
sebevarelse når sammensetningen av oljen er konstant.

I ligning 2.4 setter vi nå inn for volumhastighetenuo fra ligning 2.1 og får

∂
∂x

�
Ckxbo

µo

∂ po

∂x

�
+qo = φ

∂bo

∂ t
:

Siden vi har antatt at volumfaktorenbo kun avhenger av trykket, kan vi skrive

∂bo

∂ t
=

∂bo

∂ po

∂ po

∂ t
;

og dermed

∂
∂x

�
Ckxbo

µo

∂ po

∂x

�
+qo = φ

∂bo

∂ po

∂ po

∂ t
: . . . . . . . . . . . . . . . . . (2.5)

Dette er diffusivitetsligningen vi skal løse numerisk. Den gjelder under de forutsetnin-
ger som er gitt innledningsvis i dette kapitlet.

Ligning 2.5 uttrykker at strøm ix-retning pluss produksjon eller injeksjon er lik
ekspansjon eller kompresjon.

I den utleverteFORTRAN-koden, som løser ligning 2.5, er det brukt symboleneA2
= φ∂bo=∂ po og A9 = qo.

Enheten til leddet på høyre siden av ligningen,φ(∂bo=∂ t), er stb/rb/d. Dette er i
samsvar med enheten tilqo. Dersom en praktisk rateq0o med enhet stb/d blir produsert
fra et volumelement∆V , så blir kildeleddet i ligning 2.5 gitt vedqo = q0o=(∆V /FPB),
hvor FPB= 5.6146 [ft3/bbl] og ∆V er i ft3.

2.2 Generell diffusivitetsligning

Den generelle kontinuitetsligning er beskrevet i kompendiet til Papatzacos [3] og i
standard tekster i fluidmekanikk,

∂
∂x

(ρouo) =�
∂
∂ t

(φρo): . . . . . . . . . . . . . . . . . . . . . . . . . . (2.6)

I tillegg kan det være et kildeledd. Tettheten er gitt ved

ρo = bo(ρst
o +ρstd

g Rso=FPB); . . . . . . . . . . . . . . . . . . . . . . . . (2.7)

hvor ρst
o er oljetettheten vedstock tank forhold i lb/ft3, ρstd

g er gasstettheten vedstan-
dard forhold i lb/ft3, ogRso er oppløst gass-olje forhold i ft3/stb.

Ligning 2.7 kan utledes ved å betrakte hvordan en gitt massem fordeler seg mellom
gass- og væskefasen ved reservoarforhold og ved overflateforhold. Ved reservoarfor-
hold er all masse i oljefasen,m = ρorVor, hvor r betegner reservoarbetingelser ogV
volumet.

Merk: Modellen er for enfase strøm. For olje betyr det at trykket i reser-
voaret må være over kokepunktstrykket for oljen, ellers vil det også være
gass tilstede, og da må en bruke tofase-formulering: en konserveringslig-
ning for gassen og en for oljen.

Samme massem kan ved overflaten (indekss for “surface”) uttrykkes vedm=Vosρos+
Vgsρgs, hvor Vos er tatt vedstock tank forhold ogVgs er tatt vedstandard forhold på
overflaten. Videre erVgs = VosRso=5:61 og bo = Vos=Vor. Satt sammen så gir dette
ligning 2.7.

Vi antar nå igjen atφ er konstant og kan settes utenfor derivasjonstegnet. Setter en
inn for ρo i ligning 2.6 og foruo fra ligning 2.1, så får en

∂
∂x

�
bo(ρst

o +ρstd
g Rso=FPB)

Ckx

µo

∂ po

∂x

�
= φ

∂
∂ t

�
bo(ρst

o +ρstd
g Rso=FPB)

�
:

Sidenρst
o , ρstd

g ogRso er konstanter, reduseres dette uttrykket direkte til ligning 2.5. At
Rso er en konstant følger av at vi betrakter bare en fase, i form av olje. Siden det ikke
er fri gass tilstede i reservoaret, må trykket være over kokepunktstrykket, og da erRso

konstant.

Merk: Oppgave 1 i eksamenssettene fra desember 1995 og desember 1996
gir to eksempler på utledning av diffusivitetsligningen dersom det antas
hvordan tettheten avhenger av trykket (tilstandsligning).

2.3 Kommentar til utledning av diffusivitetsligningen

La oss betrakte volumet∆V = A∆x i Figur 2.1. Masseraten ut av volumelementet ved
x+∆x er (ρouoA)x+∆x og masseraten inn i volumelementet vedx er (ρouoA)x. Diffe-
ransen må være lik endring av masse per tidsenhet i volumelementet,

(ρouoA)x+∆x� (ρouoA)x =�
∆t(∆Vφρo)

∆t
; . . . . . . . . . . . . . . . . (2.8)

hvor ∆t betyr tidsdifferanse ved konstantx-verdi. Vi deler nå ligningen på∆V , og lar
∆x og ∆t gå mot null. Da får vi

∂
∂x

(ρouo) =�
∂
∂ t

(φρo); . . . . . . . . . . . . . . . . . . . . . . . . . . (2.9)

som i Ligning 2.6 og kan fortsette derfra. Hvordan skal vi få formulert inn en rate
i ligningen? Dette kan gjøres på to måter. Dersom vi holder oss til Ligning 2.9, så
må raten formuleres som et grensevilkår for en bestemx-verdi. For eksempel kan det
bli spesifisert at raten ved utløpet av reservoaret skal ha en angitt verdi. Denne måten
brukes generelt når en skal løse en differensialligning analytisk.

Den andre måten er å ta utgangspunkt i Ligning 2.8, altså i den diskrete formen. Vi
bruker at∆x er differanseoperator ix medt konstant. Videre så multipliserer og deler
vi med∆x på venstre siden og får,

∆x(ρouo)

∆x
∆V =�

∆t(∆Vφρo)

∆t
: . . . . . . . . . . . . . . . . . . . . . . (2.10)

Hvert ledd i denne ligningen har dimensjonen masse/tid. Inn i volumet∆V sender vi
nå en rateqi med dimensjonen masse/tid. Vi har atqi = qo∆V (ρost +Rsoρgstd), hvor
dimensjonene påqo er overflatevolum olje per tidsenhet per bulkvolum, som definert
foran. Vi innfører ogsåρo = bo(ρost +Rsoρgstd) i de to andre leddene i ligning 2.10.
Uttrykket for qi kan vi sette rett inn i Ligning 2.10, forkorte bort konstanten(ρost +
Rsoρgstd) og dele på∆V . Da får vi Ligning 2.4 i grensen at∆x og ∆t går mot null.

2.4 Diffusivitetsligning for tre faser

Uten utledning tar vi her med utvidelsen av ligning 2.5 til strøm av olje, vann og gass
i en dimensjon. Vi har da tre diffusivitetsligninger,

∂
∂x

�
Ckx

krobo

µo

∂ po

∂x

�
+qo = φ

∂
∂ t

(boSo);

∂
∂x

�
Ckx

krwbw

µw

∂ pw

∂x

�
+qw = φ

∂
∂ t

(bwSw);

∂
∂x

�
Ckx

krgbg

µg

∂ pg

∂x

�
+

∂
∂x

�
Ckx

krobo

µo
Rso

∂ po

∂x

�
+qg = φ

∂
∂ t

(bgSg +boRsoSo);

og de tre føringsbetingelsene

So +Sw +Sg = 1;

pg = po +Pcgo;

pw = po�Pcow:

Her stårS for metning,Pcgo er kapillartrykket mellom gass og olje ogPcow mellom olje
og vann. Det er også enkelt å ta med ledd som behandler olje oppløst i gassen, slik at
en kan simulere et gasskondensat felt med denneβ -formuleringen.

Funksjonssammenhengene i ligningene er slik atb;µ;Rso kun avhenger av trykket,
og de metningsavhengige parametrene avhenger av

Pcow : Pcow(Sw); kro : kro(Sw;Sg);
Pcgo : Pcgo(Sg); krw : krw(Sw);

krg : krg(Sg):

Vi skal i første omgang holde oss til ligning 2.5 med begrensningene atbo varierer
lineært med trykket gitt vedbo = borig +dbdp (po� porig), hvordbdp er konstant og
betegnerFORTRAN-symbolet for∂bo=∂ po, og indeksorig betegner startverdi. I tillegg
skal vi anta at viskositetenµo er en konstant.



Kapittel 3

Numerisk formulering

Diffusivitetsligningen kan løses analytisk kun for enkle geometrier og grensevilkår, og
med visse forenklinger. Et eksempel på dette er gitt i en av øvingene. I det alminne-
lige tilfellet må en begrense seg til numeriske løsninger, og for denne typen partielle
differensialligninger finnes det to hovedklasser av metoder:

1. Differansemetoden

2. Elementmetoden

Elementmetoden brukes ofte til simulering av statiske belastninger og sjelden til simu-
lering av strøm i et reservoar. Vi skal holde oss til differansemetoden hvor hovedpoen-
get er at alle deriverte tilnærmes med et forhold mellom differanser.

3.1 Tilnærming av deriverte

Ligning 2.5 inneholder deriverte med hensyn på både tid og avstand. I differansemeto-
den tilnærmes begge som vist under, for en vilkårlig funksjony(x) framstilt i Figur 3.1.

Den deriverte i punktet(y(xo);xo) er gitt ved

dy
dx

����
x=xo

= lim
∆x!0

y(xo +∆x=2)� y(xo)

∆x=2
;

og den tilnærmes numerisk ved hjelp av differanser. Det vil si at vi tilnærmer tangen-
tens stigningsforhold med sekantens på følgende måte

dy
dx

����
x=xo

�
y(xo +∆x=2)� y(xo�∆x=2)

∆x
=

y(x2)� y(x1)

∆x
:

Dette svarer til å ta med kun første ordens ledd i en Taylorutvikling avy(x) omkring
punktetx0.

11

tangent

sekant

y(x)

xx1 x0 x2

∆x/2 ∆x/2

Figur 3.1: Tilnærming av tangent med sekant.

3.2 Avstandsderiverte i diffusivitetsligningen

Vi skal først se hvordan en kan tilnærme deriverte med hensyn på avstandenx i lig-
ning 2.5. La oss dele det en-dimensjonale reservoaret inn i tre numeriske blokker, slik
som vist i Figur 3.2.

x

y

z

A B∆z

∆y

(x1,p1) (x2,p2) (x3,p3)

∆x1 ∆x2 ∆x3

Figur 3.2: Endimensjonalt reservoar med tre numeriske blokker.

I denne figuren betegner∆xi lengden av numerisk blokk nr.i, xi avstanden til midt-
punktet av blokki og pi trykket vedxi.

Merk: Fra og med dette punkt i framstillingen skifter vi notasjon slik at
indekso for olje sløyfes. Vær imidlertid oppmerksom på at variabelenPO
brukes til å betegne trykket i den utleverteFORTRAN-kode.

La A betegne grensen mellom blokk 1 og 2, ogB grensen mellom 2 og 3. Da kan

vi tilnærme de deriverte ved de to grensene på følgende måte

dp
dx

����
A
�

p1� p2

x1� x2
=�

p1� p2

x2� x1
;

og

dp
dx

����
B
�

p3� p2

x3� x2
:

La oss definere en del hjelpestørrelser før vi går videre i diskretiseringen av ligningen.
Vi setter

∆p+2 = p3� p2;

∆p�2 = p1� p2;

∆x+2 = x3� x2 =
1
2
(∆x2 +∆x3);

∆x�2 = x2� x1 =
1
2
(∆x1 +∆x2) = ∆x+1 ;

N+x2 =

C

�
kxb
µ

�+
2

∆x+
2

;

N�x2 =

C

�
kxb
µ

��
2

∆x�
2

:

Notasjonen er her at for en vilkårlig variabelY betegner[Y ]+2 en verdi tatt ved grense-
flaten mellom blokkene 2 og 3, eller en middelverdi mellom blokkene.

StørrelsenN+x2 representerer midlere strømningskoeffisient eller transmissibilitet
mellom blokk 2 og 3. Volumfaktorenb og viskositetenµ kan avhenge av trykket
p(x; t), som altså er en funksjon av både avstandx og tid t. Når simuleringsprogram-
met flytter seg et tidssteg∆t fra t til t +∆t, kan en vise at det oppnås god stabilitet og
mest korrekte verdier for trykkløsningen dersom det brukes tidsmidlete verdier forb
og µ i uttrykket for Nx, se avsnitt 3.4. Det vil si at en for volumfaktoren beregnet ved
grenseflatenB bruker

[b]+2 =
1
4
(b(p(x2; t +∆t))+b(p(x3; t +∆t))+b(p(x2; t))+b(p(x3; t))):

Således blir denb-verdien som inngår iNX-ledd både tids- og avstandsmidlet.

Merk: I avsnitt 3.4 blir omtalt hva som menes med implisitt formulering:
Verdien av de ukjente trykkene velges ved slutten av tidssteget. For å få en
konsistent, fullstendig implisitt formulering måb og µ i Nx også beregnes
ved slutten av tidssteget. Dette har imidlertid liten praktisk betydning siden
b og µ er svakt trykkavhengige for olje.

3.3 Diskretisering i tid og rom

Vi er nå klar til å diskretisere diffusivitetsligningen 2.5 slik at den kan løses numerisk.
I Figur 3.3 har den generelle blokken(i) grenseflatenA mot naboen(i�1) i negativ
x-retning og grenseflatenB mot naboen(i+1) i positiv x-retning.

i-1 i i+1

A B

Figur 3.3: Generell blokknummerering.

For å forenkle framstillingen setter vi hjelpestørrelsenY lik

Y =
Ckxb

µ
∂ p
∂x

;

og Ligning 2.5 skrives som

∂Y
∂x

+A9 = A2
∂ p
∂ t

: . . . . . . . . . . . . . . . . . . . . . . . . . . . . . (3.1)

Vi bruker differansemetoden til å diskretisere Ligning 3.1, høyre siden i romligx-
retning, og venstre siden i tid, og får1

Yi

��
B� Yi

��
A

∆xi
+A9i = A2i

pi(t +∆t)� pi(t)

∆t
: . . . . . . . . . . . . . . . . (3.2)

Videre tilnærmes

Yi

��
B =

�
Ckxb

µ
∂ p
∂x

�
B
;

�

�
Ckxb

µ

�
B

�
∂ p
∂x

�
B
;

�

�
Ckxb

µ

�+
i

∆p+i
∆x+

i

;

� N+xi �∆p+i ;

1For enkelthets skyld bruker vi fortsatt standard likhetstegn mellom venstre og høyre siden, selv om
disse nå er diskretiserte tilnærminger til differensialligningens høyre og venstre side.

og tilsvarende for det andre leddet,

Yi

��
A ��N�xi �∆p�i :

Vi definerer nå

O+xi =
N+xi

∆xi
;

O�xi =
N�xi

∆xi
;

som innsatt i Ligning 3.2 gir diffusivitetsligningen på følgende numerisk form

O+xi �∆p+i +O�xi �∆p�i +A9i = A2i(pi(t +∆t)� pi(t))=∆t: . . . . . (3.3)

Det er løsningen av denne simuleringsligningen som er programmert i den utdelte
FORTRAN-koden. Programmet er laget med en initialiseringsdel som gir verdi til alle
variable vedt = 0, og med en tidsstegssløyfe som beregner trykkene i de numeriske
blokkene etter et fastsatt nytt tidssteg∆t, når alle størrelsene i ligningen, inklusive
trykkene i hver blokk, er kjent på tidsnivåt.

3.4 Valg av numerisk formulering

Alle størrelser som inngår i Ligning 3.3 er kjente på tidsnivåt og simuleringsprogram-
met oppdaterer dem til nivåt +∆t. På høyre side av ligningen inngår trykket på begge
nivåene som følge av tilnærmingen av den tidsderiverte. Vi har imidlertid hittil ikke
angitt på hvilket tidsnivå leddene∆p�i skal evalueres. De tre mest vanlige valgene er
angitt i Tabell 3.1.

Numerisk formulering Tidsnivå for∆p�i
eksplisitt t
implisitt t +∆t
Crank-Nicolson t +∆t=2

Tabell 3.1: De tre vanligste numeriske formuleringer.

Før vi går videre trenger vi noen begreper fra numerisk analyse. De vil bare bli
omtalt her og interesserte kan se nærmere i boken til Carnahanet al [4] eller andre
standardverker i numeriske metoder for utfyllende kommentarer.

La oss anta atu er den korrekte, eventuelt analytiske, løsning til en differensial-
ligning, og atv er den tilsvarende numeriske løsning.Diskretiseringsfeilen er gitt ved
w = u� v. Dersomw ! 0 når(∆x;∆t)! 0, sier vi at den numeriske formuleringen

konvergerer. Dersom differensialligningen i numerisk formulering går mot den opp-
rinnelige differensialligning når(∆x;∆t)! 0, sier vi at den numeriske formuleringen
er konsistent med differensialligningen. Dersom den numeriske løsningenv er ende-
lig under hele simuleringen, sier vi at løsningen erstabil, selv om den kan fluktuere
eller oscillere mellom tidsstegene. Dersom løsningsrutinen som brukes er iterativ, se
avsnitt 3.7, brukes også begrepetkonvergens om den trinnvise tilnærming til en endelig
verdi for v innen et gitt tidssteg.

For de tre formuleringene i Tabell 3.1 kan det vises at følgende egenskaper gjelder

eksplisitt er enklest å programmere; trenger ingen løsningsrutine;w�O(∆t)+O(∆x2);
blir ustabil når∆t=∆x2 > 1=2 og er derfor betinget stabil

implisitt er ubetinget stabil og blir derfor ofte foretrukket; brukes i simuleringsmodellen
ECLIPSE; må ha en løsningsrutine;w� O(∆t)+O(∆x2)

Crank-Nicolson er også ubetinget stabil, men krever noe mer programmeringsarbeid
enn for implisitt formulering; har en tendens til å oscillere rundt den riktige
løsning;w� O(∆t2)+O(∆x2)

Merk: Disse uttrykkene for diskretiseringsfeil forutsetter at∆t og ∆x er
dimensjonsløse, det vil si at diffusivitetsligningen er gjort dimensjonsløs
før diskretiseringen.

3.5 Eksplisitt formulering

Nå velges∆p�i i Ligning 3.3 ved tidt, og er derfor kjente tall. Den eneste ukjente i
lingningen erpi(t +∆t) som vi kan finne direkte:

pi(t +∆t) =
∆t
A2i

(O+xi �∆p+i (t)+O�xi �∆p�i (t)+A9i)+ pi(t): . . . . . . (3.4)

Vi får altså ikke noe lineært ligningssystem med denne formuleringen. Nytt trykk i
blokk nummeri finnes nesten uavhengig av trykkene i naboblokkenei�1. Det er en
svak kopling via midling til naboblokkene av den trykkavhengige koeffisientenb=µ
som inngår iN�x , se avsnitt 3.2.

3.6 Implisitt formulering

Her velges∆p�i i Ligning 3.3 ved tidt +∆t.

Merk: Vi forenkler nå notasjonen slik at trykkpi(t) er ved tidt, mens
trykk pi(t +∆t) betegnes forenklet bare sompi.

Vi bruker definisjonene på∆p�i fra avsnitt 3.2, og setter inn i Ligning 3.3 og får

O+xi(pi+1� pi)+O�xi(pi�1� pi)+A9i =
A2i

∆t
(pi� pi(t));

som rearrangert gir

O�xi pi�1 +(�O�xi �O+xi �
A2i

∆t
) pi +O+xi pi+1 =�

A2i

∆t
pi(t)�A9i:

Vi definerer følgende hjelpestørrelser

Bi = O�xi +O+xi +
A2i

∆t
;

Ai = O�xi=Bi;

Ci = O+xi=Bi;

Di = (
A2i

∆t
pi(t)+A9i)=Bi;

og simuleringsligningen blir

�Ai pi�1 + pi�Ci pi+1 = Di: . . . . . . . . . . . . . . . . . . . . . . (3.5)

Ligningen gir altså et lineært ligningssett med like mange ukjente som numeriske blok-
ker, og hvor ligningen for en blokk avhenger av de ukjente i de to naboblokkene.

For å eksemplifisere velger vi et system med fire blokker, slik ati 2 f1;2;3;4g. Da
gir Ligning 3.5 følgende sett av ligninger:

0+ p1�C1 p2 = D1;
�A2p1 + p2�C2 p3 = D2;
�A3p2 + p3�C3 p4 = D3;

�A4 p3 + p4 +0= D4;

. . . . . . . . . . . . . . . . . . . . . . . . (3.6)

som kan skrives som en matriseligning0
BB@

1 �C1 0 0
�A2 1 �C2 0

0 �A3 1 �C3
0 0 �A4 1

1
CCA �

0
BB@

p1
p2
p3
p4

1
CCA=

0
BB@

D1
D2
D3
D4

1
CCA ; . . . . . . . . . (3.7)

eller
)

K �~pt = ~Dt ;

hvor koeffisientmatrisen
)

K,

)

K=

0
BB@

1 �C1 0 0
�A2 1 �C2 0

0 �A3 1 �C3
0 0 �A4 1

1
CCA ;

er entridiagonal matrise, hvor kun hoveddiagonalen og de to subdiagonalene har ledd
forskjellig fra 0.

3.7 Iterativ løsningsmetode

En ligning av typen 3.7 kan løses iterativt eller direkte, ved eliminasjon. Som vi skal
demonstrere senere, i forbindelse med en øving, er direkte metoder å foretrekke for en-
dimensjonale system. Det samme er tilfelle for to-dimensjonale problemer opp til en
viss størrelse. I tre dimensjoner derimot blir det selv for ‘små’ systemer nødvendig å
bruke iterative metoder. En moderne, industriell modell somECLIPSE, for eksempel,
har ikke direkte løsningsmetode innlagt.

La oss beskrive prinsippet for iterative metoder med et enkelt eksempel, to lignin-
ger med to ukjente:

1 : x1 +2x2 = 5;
2 : x1 +3x2 = 7;

eller

1 : x1 = 5�2x2;
2 : x2 = (7� x1)=3:

La k betegne iterasjonstelleren. Prosedyren forGauss-Jordan’s metode er: Start med
antatte verdier på bådex1 og x2. For iterasjonsrundek + 1, beregn samtidigxk+1

1 =

5�2xk
2 og xk+1

2 = (7� xk
1)=3, og forsett inntil konvergens.

Prosedyren forGauss-Seidel’s metode er litt anderledes: Beregn førstxk+1
1 = 5�

2xk
2 og deretterxk+1

2 = (7� xk+1
1 )=3. En bruker altså oppdaterte verdier straks de er

generert. Denne metoden er lagt inn som en subrutine i det utleverte programmet og
reprodusert nedenfor. Iterasjonssløyfen kallesitera , sløyfen for løsing av trykkene i
hver blokk kallespressures . Ligning 3.5 er brukt inne i denne sløyfen. Triggeren
isw brukes til å sjekke for konvergens.

Gauss-Seidel subrutine i utlevert program

subroutine gsit(mx,eps)
use felles_matriser
implicit none

integer :: kkk, i, isw, im, ip, mx, kkkcum
real(dp) :: eps

itera: do kkk = 1,300
kkkcum = kkk
isw = 0
im = 1

pressures: do i = 1,mx

if (i .gt. 1) im = i-1
if (i .lt. mx) ip = i+1

po(i) = a(i)*po(im)+c(i)*po(ip)+d(i)

if (isw == 0) then
if (dabs(comp(i)-po(i)) > eps) isw = 1

end if

comp(i) = po(i)

end do pressures

if (isw .eq. 0) exit

end do itera

write (6,"(’ kkkcum .........’, i4/)") kkkcum

return
end

3.8 Direkte løsningsmetode

La oss anskueliggjøre hvordan det lineære ligningssystemet 3.6 kan løses ved elimi-
nasjon. Ligningene tenkes nummerert fra 1 til 4 fra toppen og ned. Framgangsmåten
er å eliminerep1 fra andre ligning ved å bruke første ligning; deretterp2 fra tredje
ligning ved å bruke den oppdaterte andre ligning —hvorp1 er fjernet; og til sistp3 fra
fra fjerde ligning.

Sammenligner en med ligningen på formen 3.7, så har vi nå endret matriseelemen-

ter i
)

K slik at det kun er nuller under diagonalen. Lar vi
 betegne et matriseelement
som har fått endret verdi, har Ligning 3.7 nå følgende utseende:0

BB@
1 �C1 0 0
0 
 �C2 0
0 0 
 �C3
0 0 0 


1
CCA �

0
BB@

p1
p2
p3
p4

1
CCA=

0
BB@

D1







1
CCA :

Fjerde ligning, tilsvarende linje nummer 4 i
)

K, innholder nå barep4 som ukjent. Denne
bestemmes og settes inn i tredje ligning, som da bare harp3 som ukjent, etc. Prose-
dyren består altså i to sveip, først nedover for å skaffe nuller under diagonalen, så

oppover igjen med suksessiv innsetting av funne trykk. Denne algoritmen er lagt inn
som en subrutine i det utleverte program, og er reprodusert under. I line 04 betegner
al diagonalleddet; videre ers den endrete høyreside av ligningene.

En kan merke seg at det kun opereres på ledd i de tre diagonalene i
)

K, eller innenfor
båndet til denne båndmatrisen. Løsningsrutinentridia er derfor meget effektiv og
trenger lite lagerplass.

Subrutine for løsing av lineært ligningssett med tridiagonal koeffisientmatrise

subroutine tridia(mx)
use felles_matriser

implicit none

integer :: i, mx, k

al(1) = 1
s(1) = d(1)

! Lager øvre triangulær matrise med
! kun nuller under hoveddiagonalen.
upper_trian: do i = 2,mx

al(i) = 1. - a(i)/al(i-1)*c(i-1)
s(i) = d(i) + a(i)/al(i-1)*s(i-1)

end do upper_trian

! Løser for trykket i blokk mx
po(mx) = s(mx)/al(mx) !last pressure

! “Spaserer” oppover diagonalen igjen og løser for
! resten av trykkene.
back_subst: do k = 2,mx

i = mx - k + 1
po(i) = (s(i)+c(i)*po(i+1)) / al(i)

end do back_subst

return
end subroutine tridia

3.9 Skisse til flytskjema for utlevert program

INITIALISERING
Les data
Skriv data
Beregn konstante størrelser
Initialiser variable
BeregnOIP

α START TIDSSTEGSSLØYFE
Oppdater trykk,POLD = PO

β START YTRE ITERASJON, på ulineære ledd, dvs.b;µ
PEST = PO(1)
PO = (PO + POLD)/2
Fluidegenskaper —subrutineFLPROP
Strømningskoeffisienter —subrutineFLOCON
Løs ligningssettet med en av subrutinene
TRIDIA –dersom direkte løsning
GSIT –dersom iterativ løsning

Skriv beregnet trykk dersomIWRITE=1
BeregnE = kPO(1) - PEST k

β 0 SLUTT DersomE � ε: Gå til β ! ε er innlest toleransekriterium
Gjør materialbalanse
Skriv ut trykk og tidsstegsammendrag

α 0 SLUTT Dersom flere tidssteg: Gå tilα
STOP

program utdelt_sim
use felles_matriser
implicit none

!list of real variables in double precision:
! kx permeabilitet i x-retning, md
! nt oljevolum tilstede ved et gitt tidspunkt, stb
! ni oljevolum injisert, stb
! np oljevolum produsert, stb
! npc oljevolum produsert i %
! visorg opprinnelig viskositet, cp
! dbdp derivert av bo mhp trykk, stb/rb/psi
! delt1 de 12 første tidssteg er delt1/(1.5**(12-l)), hvor
! l er antall tidssteg, days
! eps trykktoleranse, iterasjon på ulineære ledd _og_ i
! iterativ løsningsrutine, psi
! dex delta x, ft
! dey delta y, ft
! dez delta z, ft
! fpb cubic feet per barrel, 5.61
! pi 3.14
! vol dex*dey*dez/fpb, rb
! volphi vol*phi, rb
! ckbak konstantleddet i strømningskoeffisientene
! a2 phi*dbdp
! ooip original oil in place, stb
! pest estimated value of po(1), psi
! qotot total oil rate, stb/day
! ctim kumulativ simulert tid, days
! eot total oil error
! borig original value of bo, stb/rb
! phi porosity
! porig original oil pressure, psi
! delt time step size, days
! convf conversion factor from Darcy units to oil-field
! units
! en 1.d0
! to 2.d0
!

real(dp) :: kx,nt,ni,np,npc,visorg,dbdp,delt1,eps,dez,&
dey,fpb,pi,vol,volphi,ckbak,a2,ooip,&
pest,qotot,ctim,eot,borig,phi,dex,porig,&
delt,convf,en,to

!list of integers
! stmax number of time steps
! mx number of numerical blocks
! isol specify solution method
! ipe trigger, not in use
! iskip trigger, not in use
! iwrite trigger, if equal to 1, intermediat pressures are
! written
! i loop counter
! im i-1
! ip i+1
! lcysw number of iterations per time step on non-linear terms
! kcy sum of lcysw
! l time step counter
! k loop counter
!

integer :: stmax,mx,isol,ipe,iskip,iwrite,i,im,ip,&
kcy,l,k,lcysw

open (unit=5,file=’res90dat.dat’, status=’old’)
open (unit=6,file=’res90out.dat’, status=’replace’)

! ** read and write input data

write (6,3000)
read (5,*) mx, stmax, isol, ipe, iskip, iwrite

allocate (&
a(mx),c(mx),d(mx),al(mx),s(mx),comp(mx),po(mx),&
pold(mx),oxplus(mx),oxmin(mx),a9(mx),qo(mx),bo(mx),x(mx)&

)

write (6,3010) mx, stmax, isol, ipe, iskip, iwrite
read (5,*) kx, phi, dex, porig, borig
read (5,*) visorg, dbdp, delt1, eps, dez
read (5,*) dey
write (6,3020) kx, phi, dex, porig, borig, visorg, &

dbdp, delt1, eps, dez, dey
read (5,*) (qo(i),i = 1,mx)

write (6,3030) (i,qo(i), i = 1,mx)
write (6,3080)

!
! ** begin calculations; initialize variables

!
fpb = 5.6146d0
pi = 3.1415926d0
convf = 0.00632827

en = 1.d0
to = 2.d0

oxplus(mx) = 0
oxmin(1) = 0
delt = 0
ni = 0
np = 0
ctim = 0
kcy = 0

vol = dez*dey*dex/fpb
volphi= vol*phi
ckbak = convf*kx/dex/visorg/dex
a2 = phi*dbdp

ooip = 0.0
x(1) = dex/to

pressure_points:&
do i = 2,mx

im = i -1
x(i) = dex + x(im)

end do pressure_points

write (6,3040) (i,x(i),i=1,mx)
write (6,3080)

po = porig
qo = qo /vol
a9 = qo
pold = porig

ooip_in_place:&
do i = 1,mx

ooip = ooip + volphi * borig
end do ooip_in_place

time_step_loop:&
do l = 1, stmax

pold = po

lcysw = 0
if (l <= 12) delt = delt1/1.5**(12-l)

write (6,3070)
!
! cc start iteration loop on unlinear variables i.e. bo
!

7744 continue

pest = po(1)
kcy = kcy + 1

po = (po + pold)/to

call flprop(mx,porig,dbdp,borig)
call flocon(mx,delt,a2,ckbak)

select case(isol)

case(1)
call tridia(mx)

case(2)
po = to*po - pold
comp = po
call gsit(mx,eps)

end select

if (iwrite == 1) then
write(6,3050) (i,po(i), i = 1,mx)
write(6,3080)

end if

lcysw = lcysw + 1

check_convergence_of_nonlinear_term:&
if (&

dabs(po(1)-pest) > eps &
.and. &

lcysw < 4) then

goto 7744

end if check_convergence_of_nonlinear_term

!
! start material balance
!

ctim = ctim + delt

nt = 0
qotot = 0.

call flprop(mx,porig,dbdp,borig)

matbal:&
do i = 1,mx

nt = nt + volphi * bo(i)
if (qo(i) .gt. 0.) ni = ni+qo(i)*vol*delt
if (qo(i) .lt. 0.) np = np-qo(i)*vol*delt
if (qo(i) .lt. 0.) qotot = qotot-qo(i)*vol

end do matbal

eot = (-ni+np+nt-ooip)/(ooip+ni)*100
npc = np/ooip*100

if (iwrite /= 1) then
write (6,3050) (i,po(i),i=1,mx)

end if

write (6,3060)l,ctim,np,delt,kcy,ni,nt,qotot,eot,npc

end do time_step_loop

!
! ** close files
!

close (5)
close (6)

!
! ** format statements
!

3000 format (&
’ ***** one dimensional, one phase simulator ***** ’/&
’ ------------------------ ’//&

’ basic assumptions of this model include: ’//&
’ - horizontal flow of a compressible fluid ’/&
’ - pressure dependent volume factor ’/&
’ - pressure independent viscosity ’/&
’ - homogeneous rock properties, that is the same ’/&
’ porosity and permeability in all blocks. ’/&
’ - flowing fluid is undersaturated oil ’//)

!
3010 format (&

’ number of blocks in x-direction, mx .......... :’, i3 /&
’ maximum number of time steps, stmax .......... :’, i3 /&
’ solution method used, isol ................... :’, i3 /&
’ ipe .......................................... :’, i3 /&
’ iskip ........................................ :’, i3 /&
’ write option, iwrite ......................... :’, i3 //)

!
3020 format (&

’ horizontal permeability, kx, md ............. :’, f12.4/&
’ porosity, phi, fraction ...................... :’, f8.4

/&
’ delta length (x-dir), dex, ft ................ :’, f10.2/&
’ original oil pressure, porig, psia ........... :’, f10.1/&
’ original oil fvf, borig, std vol / res vol ... :’, f10.4/&
’ original oil viscosity, visorg, cp ........... :’, f10.4/&
’ oil fvf pressure derivative, dbdp, ’, /&
’ std vol/res vol/psi ........................ :’, f10.8/&
’ time increment, delt1, days .................. :’, f10.6/&
’ pressure tolerence, eps ...................... :’, f8.5

/&
’ block width, dez, ft ......................... :’, f8.1

/&
’ block thickness, dey, ft ..................... :’, f8.1

//)
!

3030 format (&
’ block no. :’, i3, ’ oil rate, qo, stb/d .... ’, f10.1)

!
3040 format (&

’ block no. :’, i3, ’ pressure point, x(i), ft ’, f8.1)
!

3050 format (&
’ block no. :’, i3, ’ oil pressure, psia ..... ’, f12.3)

!
3060 format (2x,’steps=’,i4,t22,’time=’,d12.4,t43,’np=’,d12.4/2x,&

’delt=’,d12.4,t22,’cum cycle=’,i4,t43,’ni=’,d12.4/2x,&

’nt=’,d12.4,t22,’qotot=’,d12.4,t43,’cum oil error=’,d12.4/2x,&
’npc=’,d12.4)

!
3070 format (25(’*****’)/)

!
3080 format (/)

! stop
end

!
! ****************************************************************
! * end of main program
! ****************************************************************
!
!
! ****************************************************************
! * subroutine tridia
! ****************************************************************
!
subroutine tridia(mx)

use felles_matriser
implicit none

integer :: i, mx, k

al(1) = 1
s(1) = d(1)

upper_trian: do i = 2,mx
al(i) = 1. - a(i)/al(i-1)*c(i-1)
s(i) = d(i) + a(i)/al(i-1)*s(i-1)

end do upper_trian

po(mx) = s(mx)/al(mx) !last pressure

back_subst: do k = 2,mx
i = mx - k + 1
po(i) = (s(i)+c(i)*po(i+1)) / al(i)

end do back_subst

return
end subroutine tridia
!
! ****************************************************************
! * end of subroutine tridia
! ****************************************************************

!
!
! ****************************************************************
! * subroutine gsit
! ****************************************************************
!
subroutine gsit(mx,eps)

use felles_matriser
implicit none

integer :: kkk, i, isw, im, ip, mx, kkkcum
real(dp) :: eps

itera: do kkk = 1,300
kkkcum = kkk
isw = 0
im = 1

pressures: do i = 1,mx
if (i .gt. 1) im = i-1
if (i .lt. mx) ip = i+1

po(i) = a(i)*po(im)+c(i)*po(ip)+d(i)

if (isw == 0) then
if (dabs(comp(i)-po(i)) > eps) isw = 1

end if

comp(i) = po(i)

end do pressures

if (isw .eq. 0) exit

end do itera

write (6,"(’ kkkcum .........’, i4/)") kkkcum

return
end
!
! ****************************************************************
! * end subroutine gsit
! ****************************************************************
!

!
! ****************************************************************
! * subroutine flprop
! ****************************************************************
!
subroutine flprop(mx,porig,dbdp,borig)

use felles_matriser
implicit none

integer :: mx, i
real(dp) :: porig,dbdp,borig

bo = borig + (po - porig)*dbdp

return
end subroutine flprop
!
! ****************************************************************
! * end subroutine flprop
! ****************************************************************
!
! ****************************************************************
! * subroutine flocon
! ****************************************************************
!
subroutine flocon(mx,delt,a2,ckbak)

use felles_matriser
implicit none

integer :: mx,i,im
real(dp) :: delt,a2,ckbak,b

oxp: do i = 2, mx
im = i-1
oxmin(i) = ckbak*(bo(i)+bo(im))/2
oxplus(im) = oxmin(i)

end do oxp
acd: do i = 1, mx

b = oxmin(i)+oxplus(i)+a2/delt
a(i) = oxmin(i)/b
c(i) = oxplus(i)/b
d(i) =(a2/delt*pold(i)+a9(i))/b

end do acd
return
end subroutine flocon
!
! ****************************************************************
! * end subroutine flocon

! ****************************************************************
!

3.10 Module felles-matriser
Module felles_matriser

use nrtype
! integer, parameter :: dp = kind(1.d0)

real(dp), dimension(:), allocatable ::&
a,c,d,al,s,comp,po,& !common/blk1/
pold,oxplus,oxmin,a9,& !common/blk2/
qo,bo,x !dimensioned matrices

end module felles_matriser

3.11 Module nrtype
Module nrtype
integer, parameter :: dp = kind(1.d0)
integer, parameter :: sp = kind(1.0)
end module nrtype

3.12 Datafil
6,4,1,1,1,0 mx,stmax,isol,ipe,iskip,iwrite
100,.05,10,2000,.7889 kx,phi,dex,porig,borig
1.358,8.03d-06,.002,0.001,1 visorg,dbdp,delt1,eps,dez
10 dey
-1,4*0,1 oil flow rates (mx values)

3.13 Resultatfil
***** one dimensional, one phase simulator *****

------------------------

basic assumptions of this model include:

- horizontal flow of a compressible fluid
- pressure dependent volume factor
- pressure independent viscosity
- homogeneous rock properties, that is the same

porosity and permeability in all blocks.
- flowing fluid is undersaturated oil

number of blocks in x-direction, mx .......... : 6
maximum number of time steps, stmax .......... : 4
solution method used, isol ................... : 1
ipe .......................................... : 1
iskip ........................................ : 1
write option, iwrite ......................... : 0

horizontal permeability, kx, md ............. : 100.0000
porosity, phi, fraction ...................... : 0.0500
delta length (x-dir), dex, ft ................ : 10.00
original oil pressure, porig, psia ........... : 2000.0
original oil fvf, borig, std vol / res vol ... : 0.7889
original oil viscosity, visorg, cp ........... : 1.3580
oil fvf pressure derivative, dbdp,

std vol/res vol/psi ........................ :0.00000803
time increment, delt1, days .................. : 0.002000
pressure tolerence, eps ...................... : 0.00100
block width, dez, ft ......................... : 1.0
block thickness, dey, ft ..................... : 10.0

block no. : 1 oil rate, qo, stb/d .... -1.0
block no. : 2 oil rate, qo, stb/d .... 0.0
block no. : 3 oil rate, qo, stb/d .... 0.0
block no. : 4 oil rate, qo, stb/d .... 0.0
block no. : 5 oil rate, qo, stb/d .... 0.0
block no. : 6 oil rate, qo, stb/d .... 1.0

block no. : 1 pressure point, x(i), ft 5.0
block no. : 2 pressure point, x(i), ft 15.0
block no. : 3 pressure point, x(i), ft 25.0
block no. : 4 pressure point, x(i), ft 35.0
block no. : 5 pressure point, x(i), ft 45.0
block no. : 6 pressure point, x(i), ft 55.0

**************************************************

block no. : 1 oil pressure, psia ..... 1997.259
block no. : 2 oil pressure, psia ..... 1999.584
block no. : 3 oil pressure, psia ..... 1999.946
block no. : 4 oil pressure, psia ..... 2000.054
block no. : 5 oil pressure, psia ..... 2000.416
block no. : 6 oil pressure, psia ..... 2002.741
steps= 1 time= 0.2312D-04 np= 0.2312D-04
delt= 0.2312D-04 cum cycle= 2 ni= 0.2312D-04
nt= 0.4215D+01 qotot= 0.1000D+01 cum oil error= 0.0000D+00
npc= 0.5485D-03

**************************************************

block no. : 1 oil pressure, psia ..... 1993.878
block no. : 2 oil pressure, psia ..... 1998.504
block no. : 3 oil pressure, psia ..... 1999.729
block no. : 4 oil pressure, psia ..... 2000.271
block no. : 5 oil pressure, psia ..... 2001.496
block no. : 6 oil pressure, psia ..... 2006.122
steps= 2 time= 0.5781D-04 np= 0.5781D-04
delt= 0.3468D-04 cum cycle= 4 ni= 0.5781D-04
nt= 0.4215D+01 qotot= 0.1000D+01 cum oil error= 0.2107D-13
npc= 0.1371D-02

**************************************************

block no. : 1 oil pressure, psia ..... 1989.815
block no. : 2 oil pressure, psia ..... 1996.558
block no. : 3 oil pressure, psia ..... 1999.213
block no. : 4 oil pressure, psia ..... 2000.787
block no. : 5 oil pressure, psia ..... 2003.443
block no. : 6 oil pressure, psia ..... 2010.185
steps= 3 time= 0.1098D-03 np= 0.1098D-03
delt= 0.5202D-04 cum cycle= 6 ni= 0.1098D-03
nt= 0.4215D+01 qotot= 0.1000D+01 cum oil error= 0.2107D-13
npc= 0.2606D-02

**************************************************

block no. : 1 oil pressure, psia ..... 1985.067
block no. : 2 oil pressure, psia ..... 1993.695
block no. : 3 oil pressure, psia ..... 1998.317
block no. : 4 oil pressure, psia ..... 2001.684
block no. : 5 oil pressure, psia ..... 2006.305
block no. : 6 oil pressure, psia ..... 2014.933
steps= 4 time= 0.1879D-03 np= 0.1879D-03
delt= 0.7804D-04 cum cycle= 8 ni= 0.1879D-03
nt= 0.4215D+01 qotot= 0.1000D+01 cum oil error= 0.2107D-13
npc= 0.4457D-02


